NAG Library Routine Document

F08BAF (DGELSY)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08BAF (DGELSY) computes the minimum norm solution to a real linear least squares problem

$$\min_x \|b - Ax\|_2$$

using a complete orthogonal factorization of A. A is an m by n matrix which may be rank-deficient. Several right-hand side vectors b and solution vectors x can be handled in a single call.

2 Specification

```
SUBROUTINE F08BAF (M, N, NRHS, A, LDA, B, LDB, JPVT, RCOND, RANK, WORK,    &
                   LWORK, INFO)

INTEGER            M, N, NRHS, LDA, LDB, JPVT(*), RANK, LWORK, INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*), RCOND, WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name ***dgelsy***.

3 Description

The right-hand side vectors are stored as the columns of the m by r matrix B and the solution vectors in the n by r matrix X.

F08BAF (DGELSY) first computes a QR factorization with column pivoting

$$AP = Q \begin{pmatrix} R_{11} & R_{12} \\ 0 & R_{22} \end{pmatrix},$$

with R_{11} defined as the largest leading sub-matrix whose estimated condition number is less than $1/\mathrm{RCOND}$. The order of R_{11}, RANK, is the effective rank of A.

Then, R_{22} is considered to be negligible, and R_{12} is annihilated by orthogonal transformations from the right, arriving at the complete orthogonal factorization

$$AP = Q \begin{pmatrix} T_{11} & 0 \\ 0 & 0 \end{pmatrix} Z.$$

The minimum norm solution is then

$$X = PZ^{\mathrm{T}} \begin{pmatrix} T_{11}^{-1} Q_1^{\mathrm{T}} b \\ 0 \end{pmatrix}$$

where Q_1 consists of the first RANK columns of Q.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrices B and X.

Constraint: $\mathrm{NRHS} \geq 0$.

4: $\mathrm{A}(\mathrm{LDA},*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1,\mathrm{N})$.

On entry: the m by n matrix A.

On exit: A has been overwritten by details of its complete orthogonal factorization.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08BAF (DGELSY) is called.

Constraint: $\mathrm{LDA} \geq \max(1,\mathrm{M})$.

6: $\mathrm{B}(\mathrm{LDB},*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1,\mathrm{NRHS})$.

On entry: the m by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08BAF (DGELSY) is called.

Constraint: $\mathrm{LDB} \geq \max(1,\mathrm{M},\mathrm{N})$.

8: $\mathrm{JPVT}(*)$ – INTEGER array *Input/Output*

Note: the dimension of the array JPVT must be at least $\max(1,\mathrm{N})$.

On entry: if $\mathrm{JPVT}(i) \neq 0$, the ith column of A is permuted to the front of AP, otherwise column i is a free column.

On exit: if $\mathrm{JPVT}(i) = k$, then the ith column of AP was the kth column of A.

9: RCOND – REAL (KIND=nag_wp) *Input*

On entry: used to determine the effective rank of A, which is defined as the order of the largest leading triangular sub-matrix R_{11} in the QR factorization of A, whose estimated condition number is $< 1/\mathrm{RCOND}$.

Suggested value: if the condition number of A is not known then $\mathrm{RCOND} = \sqrt{(\epsilon)/2}$ (where ϵ is ***machine precision***, see X02AJF) is a good choice. Negative values or values less than ***machine***

precision should be avoided since this will cause A to have an effective rank $= \min(M, N)$ that could be larger than its actual rank, leading to meaningless results.

10: RANK – INTEGER *Output*

On exit: the effective rank of A, i.e., the order of the sub-matrix R_{11}. This is the same as the order of the sub-matrix T_{11} in the complete orthogonal factorization of A.

11: WORK($\max(1, \text{LWORK})$) – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO $= 0$, WORK(1) contains the minimum value of LWORK required for optimal performance.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08BAF (DGELSY) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance,

$$\text{LWORK} \geq \max(k + 2 \times \text{N} + nb \times (\text{N} + 1), 2 \times k + nb \times \text{NRHS}),$$

where $k = \min(\text{M}, \text{N})$ and nb is the optimal ***block size***.

Constraint: $\text{LWORK} \geq k + \max(2 \times k, \text{N} + 1, k + \text{NRHS})$, where $k = \min(\text{M}, \text{N})$ or LWORK $= -1$.

13: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

See Section 4.5 of Anderson *et al.* (1999) for details of error bounds.

8 Parallelism and Performance

F08BAF (DGELSY) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08BAF (DGELSY) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The complex analogue of this routine is F08BNF (ZGELSY).

10 Example

This example solves the linear least squares problem

$$\min_x \|b - Ax\|_2$$

for the solution, x, of minimum norm, where

$$A = \begin{pmatrix} -0.09 & 0.14 & -0.46 & 0.68 & 1.29 \\ -1.56 & 0.20 & 0.29 & 1.09 & 0.51 \\ -1.48 & -0.43 & 0.89 & -0.71 & -0.96 \\ -1.09 & 0.84 & 0.77 & 2.11 & -1.27 \\ 0.08 & 0.55 & -1.13 & 0.14 & 1.74 \\ -1.59 & -0.72 & 1.06 & 1.24 & 0.34 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 7.4 \\ 4.2 \\ -8.3 \\ 1.8 \\ 8.6 \\ 2.1 \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

10.1 Program Text

```
    Program f08bafe

!     F08BAF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgelsy, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nb = 64, nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: rcond
      Integer                          :: i, info, lda, lwork, m, n, rank
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), b(:), work(:)
      Integer, Allocatable             :: jpvt(:)
!     .. Executable Statements ..
      Write (nout,*) 'F08BAF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      lwork = 3*n + nb*(n+1)
      Allocate (a(lda,n),b(m),work(lwork),jpvt(n))

!     Read A and B from data file

      Read (nin,*)(a(i,1:n),i=1,m)
      Read (nin,*) b(1:m)

!     Initialize JPVT to be zero so that all columns are free

      jpvt(1:n) = 0

!     Choose RCOND to reflect the relative accuracy of the input data

      rcond = 0.01_nag_wp

!     Solve the least squares problem min( norm2(b - Ax) ) for the x
!     of minimum norm.

!     The NAG name equivalent of dgelsy is f08baf
      Call dgelsy(m,n,1,a,lda,b,m,jpvt,rcond,rank,work,lwork,info)
```

```
!     Print solution

      Write (nout,*) 'Least squares solution'
      Write (nout,99999) b(1:n)

!     Print the effective rank of A

      Write (nout,*)
      Write (nout,*) 'Tolerance used to estimate the rank of A'
      Write (nout,99998) rcond
      Write (nout,*) 'Estimated rank of A'
      Write (nout,99997) rank

99999 Format (1X,7F11.4)
99998 Format (3X,1P,E11.2)
99997 Format (1X,I6)
    End Program f08bafe
```

10.2 Program Data

```
F08BAF Example Program Data

  6      5                        :Values of M and N

 -0.09   0.14  -0.46   0.68   1.29
 -1.56   0.20   0.29   1.09   0.51
 -1.48  -0.43   0.89  -0.71  -0.96
 -1.09   0.84   0.77   2.11  -1.27
  0.08   0.55  -1.13   0.14   1.74
 -1.59  -0.72   1.06   1.24   0.34 :End of matrix A

  7.4
  4.2
 -8.3
  1.8
  8.6
  2.1                             :End of vector b
```

10.3 Program Results

```
 F08BAF Example Program Results

 Least squares solution
      0.6344     0.9699    -1.4402     3.3678     3.3992

 Tolerance used to estimate the rank of A
      1.00E-02
 Estimated rank of A
      4
```